#
# This file is part of the GROMACS molecular simulation package.
#
# Copyright 2024- The GROMACS Authors
# and the project initiators Erik Lindahl, Berk Hess and David van der Spoel.
# Consult the AUTHORS/COPYING files and https://www.gromacs.org for details.
#
# GROMACS is free software; you can redistribute it and/or
# modify it under the terms of the GNU Lesser General Public License
# as published by the Free Software Foundation; either version 2.1
# of the License, or (at your option) any later version.
#
# GROMACS is distributed in the hope that it will be useful,
# but WITHOUT ANY WARRANTY; without even the implied warranty of
# MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE.  See the GNU
# Lesser General Public License for more details.
#
# You should have received a copy of the GNU Lesser General Public
# License along with GROMACS; if not, see
# https://www.gnu.org/licenses, or write to the Free Software Foundation,
# Inc., 51 Franklin Street, Fifth Floor, Boston, MA  02110-1301  USA.
#
# If you want to redistribute modifications to GROMACS, please
# consider that scientific software is very special. Version
# control is crucial - bugs must be traceable. We will be happy to
# consider code for inclusion in the official distribution, but
# derived work must not be called official GROMACS. Details are found
# in the README & COPYING files - if they are missing, get the
# official version at https://www.gromacs.org.
#
# To help us fund GROMACS development, we humbly ask that you cite
# the research papers on the package. Check out https://www.gromacs.org.

set(GMX_HDF5_MINIMUM_REQUIRED_VERSION "1.10.7")

macro(gmx_manage_hdf5)
    # Find an external hdf5 library.
    if(GMX_USE_HDF5 OR NOT DEFINED GMX_USE_HDF5)
        if(HDF5_ALREADY_SEARCHED)
            set(HDF5_FIND_QUIETLY ON)
        endif()
        find_package(HDF5 ${GMX_HDF5_MINIMUM_REQUIRED_VERSION} COMPONENTS C)
        set(HDF5_ALREADY_SEARCHED TRUE CACHE INTERNAL "True if a search for HDF5 has already been done")
    else()
        set(HDF5_FOUND FALSE FORCE)  # Don't search for HDF5 if user set GMX_USE_HDF5 to OFF
    endif()

    set(GMX_USE_HDF5 ${HDF5_FOUND} CACHE BOOL "Build GROMACS with HDF5 support (needed for handling files in H5MD format)")
    if(GMX_USE_HDF5)
        if(NOT HDF5_FOUND)
            message(FATAL_ERROR "Cannot build with HDF5 support since HDF5 was not found. Disable HDF5 by setting the option GMX_USE_HDF5=OFF.")
        endif()
    endif()

    # Mark HDF5-related CMake options as "advanced"
    get_cmake_property(vars VARIABLES)
    foreach (varname ${vars})
        if (varname MATCHES "^HDF5_C_LIBRARY_.*")
            mark_as_advanced(${varname})
        endif()
    endforeach()
endmacro()
